
#
# Using the fastest sequential code I cam up with
# with
#
  restart;
  nelems:= 100000000:
  n:= 374894756873546859847556:
  #n:= 700;
  A:= Array(1 .. 4, 1 .. nelems):
  st:= time[real]():
  A[1,..]:=  Array( 1..nelems,
                    i->i^10*n
                  ):
  A[2,..]:= length~(A[1, ..])-1:
  A[3,..]:= zip( (i,j)-> iquo(i, 10^(j-2)),
                 A[1,..],
                 A[2,..]
               ):
  A[4,..]:= map(irem, A[1,..], 1000):
  time[real]() - st;
  A:
  A[1, -2];
  A[2, -2];
  A[3, -2];
  A[4, -2];

1617.026
3748947193840728595192309[...54 digits...]0432748097025990859847556

103
374
556

  restart;
#
# Parallelize the above across numcpus tasks
#
  nelems:= 100000000:
  n:= 374894756873546859847556:
  st:= time():
#
# Procedure which is executed in each of the
# parallel tasks
#
  doTask:= proc( sv, nv)
                 local r:= nv-sv+1,
                       B:= Array(1..4,1..nv-sv+1),
                      i :
                 B[1,..]:= Array(1..r,i->(sv-1+i)^10*n):
                 B[2,..]:= length~(B[1, ..])-~1:
                 B[3,..]:= zip( (i,j)-> iquo(i, 10^(j-2)),
                                B[1,..],
                                B[2,..]
                              ): 
                 B[4,..]:= map(irem, B[1,..], 1000):
                 return B;
           end proc:
#
# Procedure which is executed after all parallel
# tasks are complete to "assemble" all of their
# results into a single output
#
  contTask:= proc( )
                  return ArrayTools:-Concatenate(2, _passed ):
             end proc:
#
# Procedure which figures out how many parallel tasks
# should be run, what the arguments for these tasks
# should be, and what should be run when all parallel
# tasks have completed
#                  
  setupTask:= proc(n, nc)
                   local i, rng:=floor(n/nc):

                   Threads:-Task:-Continue
                   ( contTask,
                     seq( Task=[doTask, (i-1)*rng+1, i*rng ],
                          i=1..nc-1
                        ),
                          Task=[doTask, (nc-1)*rng+1, n ]
                   );

              end proc:

  st:= time[real]():
  A:= Threads:-Task:-Start( setupTask,
                            nelems,
                            kernelopts(numcpus)
                          ):
  time[real]() - st;
  A[1, -2];

  A[2, -2];
  A[3, -2];
  A[4, -2];

430.110
3748947193840728595192309[...54 digits...]0432748097025990859847556

103
374
556

interface(rtablesize=10):


